*===============================================================================
* Do-file creates "Table A2: Administrative Dataset: Violence and M-Paisa Use, using Different Radii"
*===============================================================================
clear 
set more off

cd "~/Dropbox/VFD/REStat_Native_Files/Stata_Files"
local output "~/Dropbox/VFD/REStat_Native_Files/Stata_Files/Output"



use "monthpanel_analysis.dta", clear
keep if salary==1
gen trimmedobs=0
sum balance, d
replace trimmedobs =1 if (balance <=r(p99) & balance>=r(p1))
replace trimmedobs =. if balance ==.
bys identityid: egen trimmed_extusers = min(trimmedobs)
keep if trimmed_extusers==1



foreach var in violence df ied idf {
	gen d_r10_`var'=(d_`var'_10km==1 & d_`var'_5km==0)
}

label var d_r10_violence "Violent Event in (5,10] km (=1)"
label var d_r10_df "Direct Fire in (5,10] km (=1)"
label var d_r10_ied "IED Explode in (5,10] km (=1)"
label var d_r10_idf "Indirect Fire in (5,10] km (=1)"


* Table A2, Panel A: Main violence
foreach y in balance numtxns numwithdrawal numdeposit numtransferdebit { 
		reghdfe `y' d_violence_5km d_r10_violence, absorb(ym identityid c.ym#district) cluster(district)
		est store `y'
		estadd ysumm, mean prefix(in_sample_)
		unique identityid if e(sample)
		estadd local nID = r(unique)
		estadd local samplestr "Salary Users"
		estadd local monthFE "YES"
		estadd local distFE "YES"
		estadd local individFE "YES"
	}
	estout balance numtxns numwithdrawal numdeposit numtransferdebit ///
	using "`output'/table_a2a.tex", cells(b(fmt(%9.3f) star) se(par fmt(%9.3f))) ///
	starlevels(* .1 ** .05 *** .01) style(tex) keep(d_violence_5km d_r10_violence) mlabels(, dep) collabels(,none) ///
	stats(samplestr in_sample_mean nID N_clust N r2 monthFE individFE distFE, fmt(0 3 0 0 0 3 0 0 0) ///
	label("Sample" "Mean Dep Var" "\# Individuals" "\# Clusters" "\# Observations" "R-Squared" "Month-Year FE" "Individual FE" "District Trends")) ///
	label title("Administrative Dataset: Violence and M-Paisa Use") replace

	
* Table A2, Panel B: Main violence
foreach y in balance numtxns numwithdrawal numdeposit numtransferdebit { 
		reghdfe `y' d_df_5km d_r10_df d_idf_5km d_r10_idf d_ied_5km d_r10_ied , absorb(ym identityid c.ym#district) cluster(district)
		est store `y'
		estadd ysumm, mean prefix(in_sample_)
		unique identityid if e(sample)
		estadd local nID = r(unique)
		estadd local samplestr "Salary Users"
		estadd local monthFE "YES"
		estadd local distFE "YES"
		estadd local individFE "YES"
	}
	estout balance numtxns numwithdrawal numdeposit numtransferdebit ///
	using "`output'/table_a2b.tex", cells(b(fmt(%9.3f) star) se(par fmt(%9.3f))) ///
	starlevels(* .1 ** .05 *** .01) style(tex) keep(d_df_5km d_r10_df d_idf_5km d_r10_idf d_ied_5km d_r10_ied ) mlabels(, dep) collabels(,none) ///
	stats(samplestr in_sample_mean nID N_clust N r2 monthFE individFE distFE, fmt(0 3 0 0 0 3 0 0 0) ///
	label("Sample" "Mean Dep Var" "\# Individuals" "\# Clusters" "\# Observations" "R-Squared" "Month-Year FE" "Individual FE" "District Trends")) ///
	label title("Administrative Dataset: Violence and M-Paisa Use") replace

